APPENDIX:  SIMULATION CODE

*simulate 30 binary confounders;

%macro sim(num);

data Data (keep = id eplus p anycvrisk x1-x30 mergevar);

  call streaminit(123456 + &num);

  do id = 1 to 10000;

    mergevar=&num;

**************Probability of having any CV risk = .58 for Table 1 & .80 for Table 2 ;

    p_anycvrisk=0.58; 

    anycvrisk=rand("BERNOULLI", p_anycvrisk);

    array cv1[30] x1-x30;

***************Use %'s of controls with each risk factor from Table 1 of Graham's paper;


    array p1[30] _temporary_ (0.01 0.01 0.01 0.01 0.01 0.01 0.01 0.20 0.02 0.01

                              0.03 0.22 0.14 0.04 0.12 0.19 0.07 0.08 0.01 0.21

                              0.08 0.01 0.22 0.16 0.03 0.01 0.02 0.01 0.02 0.02);

    do i=1 to 30; 

      cv1[i]=0;

    end;

    if anycvrisk=1 then do;

      do i=1 to 30; 


     cv1[i]=rand("BERNOULLI",p1[i]/p_anycvrisk); 


      end;

    end;

***************Set Pr(E+)=p.  Use p0={0.001, 0.01, 0.1} ; 

    p0 = 0.1;

    p = p0 + 0.1*(1/30)*(sum(x1,x2,x3,x4,x5,x6,x7,x8,x9,x10,x11,x12,x13,x14,x15,

        x16,x17,x18,x19,x20,x21,x22,x23,x24,x25,x26,x27,x28,x29,x30));

    eplus=rand("BERNOULLI", p);

    output;

  end;

run;

*****************Set Betas;

data Beta;

*****************Pr(D+ |Beta0);

  pr_d = 0.1;

  alpha0=log(pr_d/(1 - pr_d));

*****************OR(E+);



alpha1=log(2.0);

*****************Betas for risk factors;



array b[30] b1-b30;

       do i=1 to 10;



  b[i]=log(1.25);

       end;

       do i=11 to 20;



  b[i]=log(1.50);

       end;

       do i=21 to 30;



  b[i]=log(2.0);

       end;

mergevar=&num;


run;

*****************Merge Data set with Parameter Estimates;

data ALLdata;

 merge Beta Data;

 by mergevar;

*****************Generate Y;

lp=sum(alpha0,alpha1*eplus,B1*x1,B2*x2,B3*x3,B4*x4,B5*x5,B6*x6,B7*x7,B8*x8,B9*x9,B10*x10,


B11*x11,B12*x12,B13*x13,B14*x14,B15*x15,B16*x16,B17*x17,B18*x18,B19*x19,B20*x20,


B21*x21,B22*x22,B23*x23,B24*x24,B25*x25,B26*x26,B27*x27,B28*x28,B29*x29,B30*x30); 

Z= ranuni(867+&num);

pry=exp(lp)/(1 + exp(lp));

y=0;

if Z < pry then y = 1;

run;

****************Get Propensity Score;

ods select none;

ods output ParameterEstimates=ParmEstim;

proc logistic data=Alldata desc;

model eplus = x1-x30;

run;

ods select all;

quit;

***********Keep only Parm Estimates for Variables used to calculate Propensity score;

data Parmestim (keep= Variable estimate);

length variable $20.;

set ParmEstim;

Variable="ES" || Variable;

run;

proc transpose data=Parmestim out=trParmestim;

id variable;

var estimate;

run;

data trParmestim (drop=_NAME_);

set trParmestim;

mergevar=&num;

run;

data Alldata;

 merge trParmEstim Alldata;

 by mergevar;

 lp_ps= sum(ESIntercept,ESx1*x1,ESx2*x2,ESx3*x3,ESx4*x4,ESx5*x5,ESx6*x6,ESx7*x7,ESx8*x8,ESx9*x9,ESx10*x10,



ESx11*x11,ESx12*x12,ESx13*x13,ESx14*x14,ESx15*x15,ESx16*x16,ESx17*x17,ESx18*x18,ESx19*x19,ESx20*x20,



ESx21*x21,ESx22*x22,ESx23*x23,ESx24*x24,ESx25*x25,ESx26*x26,ESx27*x27,ESx28*x28,ESx29*x29,ESx30*x30);

prlp_ps = exp(lp_ps)/(1 + exp(lp_ps));

run;

proc rank data=Alldata out=Alldata groups=5;

  var prlp_ps;

  ranks ps_quintiles;

run;

data Alldata;

set alldata;

propensityscore=ps_quintiles +1;

run;

**************Compute Risk Score;

**************subset to UNExposed observations;

data Unexposed;

set Alldata;

if eplus=0;

run;

ods select none;

ods output ParameterEstimates=RSParmEstim;

proc logistic data=Unexposed desc;

model y = x1-x30;

run;

ods select all;

quit;

***********Keep only Parm Estimates for Variables used to calculate risk score;

data RSParmestim (keep= Variable estimate);

length variable $20.;

set RSParmEstim;

Variable="RSestim" || Variable;

run;

proc transpose data=RSParmestim out=trRSParmestim;

id variable;

var estimate;

run;

data trRSParmestim (drop=_NAME_);

set trRSParmestim;

mergevar=&num;

run;

data alldata;

 merge trRSParmEstim Alldata;

 by mergevar;



lp_riskscore= sum(RSestimIntercept,RSestimx1*x1,RSestimx2*x2,RSestimx3*x3,RSestimx4*x4,



RSestimx5*x5,RSestimx6*x6,RSestimx7*x7,RSestimx8*x8,RSestimx9*x9,RSestimx10*x10,



RSestimx11*x11,RSestimx12*x12,RSestimx13*x13,RSestimx14*x14,RSestimx15*x15,RSestimx16*x16,



RSestimx17*x17,RSestimx18*x18,RSestimx19*x19,RSestimx20*x20,



RSestimx21*x21,RSestimx22*x22,RSestimx23*x23,RSestimx24*x24,



RSestimx25*x25,RSestimx26*x26,RSestimx27*x27,RSestimx28*x28,RSestimx29*x29,RSestimx30*x30);



prlp_riskscore =exp(lp_riskscore)/(1 + exp(lp_riskscore));

run;

proc rank data=alldata out=riskytiles groups=4;

  where anycvrisk=1;

  var prlp_riskscore;

  ranks riskscore;

run;

data riskytiles;

  set riskytiles;

  riskscore = riskscore + 1;

run;

data anycvrisk0;

  set alldata;

  if anycvrisk=0;

  riskscore=0;

run;

data alldata;

  set riskytiles anycvrisk0;

run;

proc sort; by riskscore; run;

proc means n min mean max;

  by riskscore;

  var prlp_riskscore;

run;

data;

  set alldata;

  sumx=0;

  sumx = sum( of x1-x30);

run;

proc freq;

  tables sumx eplus;

run;

********************Fit Logistic Regression Model;

ods select none;

ods output ParameterEstimates=LogRegModParmEstim;

proc logistic data=Alldata desc;

***class x1-x30 / ref=first; *set 0 as reference category;

model y =eplus x1-x30;

run;

ods select all;

quit;

********************Fit Risk Score Logistic Regression Model;

ods select none;

ods output ParameterEstimates=RiskScoreModParmEstim;

proc logistic data=Alldata desc;

class riskscore / ref=first; *set 0 as reference category;

model y =eplus riskscore;

run;

ods select all;

quit;

********************Fit Propensity Score Logistic Regression Model;

ods select none;

ods output ParameterEstimates=PScoreModParmEstim;

proc logistic data=Alldata desc;

class propensityscore / ref=first; *set 0 as reference category;

model y =eplus propensityscore;

run;

ods select all;

quit;

data LogRegModBias (keep=LogModEstimate LogModStdErr mergevar);

set LogRegModParmEstim;

if Variable="eplus";

rename Estimate=LogModEstimate;

rename StdErr=LogModStdErr;

mergevar=1;

run;

data RiskScoreModBias (keep=RiskScoreModEstimate RiskScoreModStdErr mergevar);

set RiskScoreModParmEstim;

if Variable="eplus";

rename Estimate=RiskScoreModEstimate;

rename StdErr=RiskScoreModStdErr;

mergevar=1;

run;

data PScoreModBias (keep=PScoreModEstimate PScoreModStdErr mergevar);

set PScoreModParmEstim;

if Variable="eplus";

rename Estimate=PScoreModEstimate;

rename StdErr=PScoreModStdErr;

mergevar=1;

run;

************Add the number of events to data set;

ods output summary=SUMy;

proc means data=alldata sum;

var y;

run;

quit;

data Sumy;

set sumy;

mergevar=1;

run;

************Get both model results for eplus in 1 data set;

************Calculate bias & percent bias; 

data EPLUS&num;

merge LogRegModBias RiskScoreModBias PScoreModBias Sumy;

by mergevar;

trueor=2.0;

ORLog=exp(LogModEstimate);

ORRS=exp(RiskScoreModEstimate);

ORPS=exp(PScoreModEstimate);

BiasLog=exp(LogModEstimate) -trueor;

BiasRS=exp(RiskScoreModEstimate) -trueor;

BiasPS=exp(PScoreModEstimate) -trueor;

PerBlog=(BiasLog/trueor)*100;

PerBRS=(BiasRS/trueor)*100;

PerBPS=(BiasPS/trueor)*100;

*************Compute Coverage Probability;

lbLog=exp(LogModEstimate - 1.96*LogModStdErr);

ubLog=exp(LogModEstimate + 1.96*LogModStdErr);



lbRS=exp(RiskScoreModEstimate - 1.96*RiskScoreModStdErr);



ubRS=exp(RiskScoreModEstimate + 1.96*RiskScoreModStdErr);




lbPS=exp(PScoreModEstimate - 1.96*PScoreModStdErr);




ubPS=exp(PScoreModEstimate + 1.96*PScoreModStdErr);

run;

data EPLUS&num;

set EPLUS&num;

*************Create counter; 

if lbLog<=trueor<=ubLog then coverprob_log=1;


else coverprob_log=0;

if lbRS<=trueor<=ubRS then coverprob_RS=1;


else coverprob_RS=0;

if lbPS<=trueor<=ubPS then coverprob_PS=1;


else coverprob_PS=0;

************Compute Empirical Power;

if abs(LogModEstimate/LogModStdErr) ge 1.96 then emppower_log= 1;



else emppower_log= 0;

if abs(RiskScoreModEstimate/RiskScoreModStdErr) ge 1.96 then emppower_RS= 1;



else emppower_RS= 0;

if abs(PScoreModEstimate/PScoreModStdErr) ge 1.96 then emppower_PS= 1;



else emppower_PS= 0;

run;

%mend sim;

%sim(1);%sim(2);%sim(3);%sim(4);%sim(5);%sim(6);%sim(7);%sim(8);%sim(9);%sim(10);

%sim(11);%sim(12);%sim(13);%sim(14);%sim(15);%sim(16);%sim(17);%sim(18);%sim(19);%sim(20);

...

%sim(991);%sim(992);%sim(993);%sim(994);%sim(995);%sim(996);%sim(997);%sim(998);%sim(999);%sim(1000);

data ALLEplus;

 set EPLUS1 EPLUS2 EPLUS3  EPLUS4 EPLUS5 EPLUS6 EPLUS7 EPLUS8 EPLUS9 EPLUS10

EPLUS11 EPLUS12 EPLUS13 EPLUS14 EPLUS15 EPLUS16 EPLUS17 EPLUS18 EPLUS19 EPLUS20

...

EPLUS991 EPLUS992 EPLUS993 EPLUS994 EPLUS995 EPLUS996 EPLUS997 EPLUS998 EPLUS999 EPLUS1000

;

 run;

ods output summary=SUMMARYLOG;

proc means data= AllEPLUS mean std min median max ;

var ORLog LogModEstimate LogModStdErr BiasLog PerBlog y_Sum; 

run;

ods output summary=SUMMARYRS;

proc means data= AllEPLUS mean std min median max ;

var ORRS RiskScoreModEstimate RiskScoreModStdErr BiasRS PerBrs ; run;

ods output summary=SUMMARYPS;

proc means data= AllEplus mean std min median max ;

var ORPS PScoreModEstimate PScoreModStdErr BiasPS PerBps ; run;

ods output summary=SUMCounter;

proc means data= AllEPLUS sum ;

var coverprob_log coverprob_RS coverprob_PS

emppower_log emppower_RS emppower_PS;

run;

proc transpose data=summarylog out=trsummarylog;

run;

proc transpose data=summaryRS out=trsummaryRS;

run;

proc transpose data=summaryPS out=trsummaryPS;

run;

proc transpose data=sumcounter out=trsumcounter;

run;

ods select all;

ods listing close;

ods rtf file = '/SimulationResults.rtf';

proc print data = trsummarylog;

title "Logistic Regression Model: descriptive statistics for bias and percent bias";

footnote "Pr(D+ |Beta0)=0.1, Pr(E+)= 0.1, OR(E+)=2";

run;

proc print data = trsummaryRS;

title "Risk Score: descriptive statistics for bias and percent bias";

run;

proc print data = trsummaryPS;

title "Propensity Score: descriptive statistics for bias and percent bias";

run;

proc print data=trSUMcounter;

title "Coverage probability and Empirical power counters";

run;

ods rtf close;

ods listing;

